
Erik Parner 

18 August 18, 2016 

Standard analysis 3-2 

A standard analysis of the agreement between two methods or measurements. The data on lung 

measurements (PEFR) were used in the lectures on Day 3. 

Statistical methods 

95% Limits-of-agreement between Wright and Mini- Wright were computed using a paired-normal model. 

The assumptions of variance homogeneity and mean difference homogeneity were assessed using a Bland-

Altman plot and normality of the difference by a Q-Q plot. Estimates of the with-in person variation and 

between-person variation of the Wright measurement was done in a Mixed-effects linear regression with 

subject as a latent factor. Estimates are presented with 95% confidence intervals. 

Results and conclusion 

Limites-of-agreemtent between Wright and Mini-Write was (-78.1;73.9) l/min. The repeatability coefficient 

for the Wright measurement was 42.4 (95% CI: 30.3-59.4) l/min. The standard deviation of with-in subjects 

Wright measurement was 15.3 (95% CI: 10.9-21.4) and between-subjects standard deviation was 117.0 

(95% CI: 82.5-165.9).  

 

Do file 
 
********************************************************************** 
* Standard3-2.do 
* Task: A standard analysis of agreement between two methods or  
*       measurements. The data on lung measurement (PEFR) were used  
*       in the lectures on Day 3. 
* Erik Parner: 18-8-2016. 
********************************************************************** 
 
graph drop _all 
 
cd "D:\Teaching\BasicBiostat\Exercises" 
 
capture log close 
log using Standard3-2.log , text replace 
 
use pefrwright, clear 
 
* Comparing Wright and Mini Wright. 
generate dif=pefrW1-pefrMW1 
generate avg=(pefrW1+pefrMW1)/2 
label var dif "Wright 1 - Mini Wright 1" 
label var avg "Average Wright 1, Mini Wright 1" 

set scheme lean1 
scatter dif avg, yline(0) name(p1,replace) 
qnorm dif,  name(p2,replace) 
graph combine p1 p2 
* Limits of agreement. 
centile dif , centile(2.5 97.5) meansd 
drop dif avg 
 
* Two measurement on Wright. 
generate dif=pefrW1-pefrW2 
generate avg=(pefrW1+pefrW2)/2 
label var dif "Wright 1 - Wright 2" 
label var avg "Average Wright 1 and Wright 2" 
scatter dif avg, yline(0) name(p1,replace) 
qnorm dif,  name(p2,replace) 
graph combine p1 p2 
drop dif 
 
* By hand computation. 
* Estimating sd of the difference under a mean=0 assumption. 
generate squardif=(pefrW1-pefrW2)^2 
sum squardif 
* Sd of difference. 
disp sqrt(r(mean)) 
disp 1.96*sqrt(r(mean)) 
* Standard variation within a person (measurement error). 
disp sqrt(r(mean))/sqrt(2) 
 
* Mixed-effects linear regression. 
qnorm avg 
* We reshape the data to have it on a long format. 
keep subject pefrW1 pefrW2 
reshape long pefrW, i(subject) j(mnumber) 
mixed pefr || subject: , reml stddeviations 
* Limits-of-agreement/Repeatabilty coefficient. 
disp 1.96*sqrt(2)*15.30667 "  95% CI: (" 1.96*sqrt(2)*10.93706  ";" 1.96*sqrt(2)*21.42202 ")" 
 
log close 
 


